

Semejanzas del plano

Adela Jaime Pastor y
 Angel Gutiérrez Rodríguez

RESUMEN

El programa cuyo listado aparece a continuación realiza traslaciones, giros, simetrías, simetrías en deslizamiento y homotecias de polígonos. Esta construido en Basic para un Apple II.

Presentamos un programa cuyo objetivo es mostrar gráficamente las semejanzas del plano. Este programa es utilizado en la Escuela de Magisterio de Valencia como complemento del tema que dedicamos al grupo de las semejanzas del plano en las clases de primer curso.

Como en la pizarra no es posible dibujar el movimiento de las figuras y las manipulaciones con objetos tienen el inconveniente de la poca habilidad de quien los mueve, pensamos que el ordenador podría ayudar, aunque también tiene sus limitaciones; la más acusada es la aparición de error si la figura "se sale" de la pantalla.

El programa está escrito en Basic para un Apple II de 64K; no obstante, se puede utilizar con cualquier capacidad de memoria, pues los gráficos son dibujados sobre la página 1. Se utilizan polígonos con un máximo de 8 vértices, por lo cual no es necesario dimensionar los vectores usados.

El fundamento del programa consiste en, elegidos un polígono (coordenadas (X,Y)) y una transformación, realizar esa transformación del polígono mediante una serie de pasos intermedios

que lo llevan desde su posición inicial a la final (ver las figuras). Cada uno de los polígonos intermedios se borra (coordenadas (M,N)) inmediatamente antes de que aparezca el siguiente (coordenadas (Z,T)), para favorecer la sensación de movimiento; esta sensación es más o menos acusada en función de la forma y tamaño del polígono, de los parámetros de la transformación y de la cantidad de pasos intermedios (que se puede guardar mediante las sentencias 990, 1140, 1440, 1800 y 1940). A veces, resulta más interesante dibujar un segmento o solamente un punto para observar su movimiento.

Esquemáticamente, la forma de actuar el programa es la siguiente (los números indican líneas del programa):


- A) Selección de la transformación que se desea realizar (10 - 220).
- B) Dibujo de los ejes de coordenadas (240 - 350).
- C) Definición, vértice a vértice, de la figura que se va a transformar (380 - 650). Se aceptan de 1 a 8 vértices.
- D) Realización de la transformación que se ha seleccionado (940 - 2000), mediante:
 - D1) Definición de las características de la transformación (vector, centro, eje, razón ...).
 - D2) Cálculo de las coordenadas para la posición final de la figura por la transformación (aplicando

las fórmulas que indicamos más adelante).


- D3) Transformación de la figura en varios pasos. Para cada paso, se calculan las coordenadas de esa posición intermedia, se borra la posición anterior y se dibuja la nueva posición (820 - 920).
- E) Después de llegar a la posición final de la transformación, se vuelve a dibujar la figura original (680 - 740).
- F) Si se desea seguir transformando la misma figura, se vuelve a A) (750 - 790) y se realiza otra vez el proceso, excepto el paso C),

que se salta (360). En caso contrario, el programa termina.


En los casos de la simetría y de la simetría en deslizamiento, hemos imitado el movimiento de "volver la hoja del libro" y, cuando la homotecia es de razón negativa, la figura hace el recorrido mediante una disminución de tamaño y un aumento posterior. Las figuras que se incluyen son copias de pantalla (hard copy) de los procesos seguidos por cada transformación. La pérdida de perpendicularidad que se observa en algunas figuras es causada por la impresora, que "alarga" la pantalla.




TRASLACION de vector $(-120, -50)$




GIRO de centro $(-10, 10)$ y ángulo -240°



SIMETRIA cuyo eje pasa por los puntos $(60, 0)$ y $(0, 40)$



SIMETRIA EN DESLIZAMIENTO con vector de traslación $(-45, -90)$ y eje de simetría que pasa por los puntos $(20, 0)$ y $(0, -40)$



HOMOTECIA de centro (40,20) y razón -1'5

Desde el punto de vista matemático, las fórmulas empleadas son:

TRASLACION DE VECTOR (A,B):

$$\left. \begin{aligned} x' &= x+rA \\ y' &= y+rB \end{aligned} \right\} \text{ con } 0 \leq r \leq 1$$

GIRO DE CENTRO (A,B) y ϕ RADIANTES:

$$\begin{aligned} x' &= x \cos(r\phi) - y \sin(r\phi) + A(1 - \cos(r\phi)) + B \sin(r\phi) \\ y' &= x \sin(r\phi) + y \cos(r\phi) + B(1 - \cos(r\phi)) - A \sin(r\phi) \end{aligned}$$

con $0 \leq r \leq 1$

SIMETRIA DE EJE $Ax + By + C = 0$:

$$p = x - 2A \frac{Ax + By + C}{A^2 + B^2} \quad q = y - 2B \frac{Ax + By + C}{A^2 + B^2}$$

$$\left. \begin{aligned} x' &= x+r(p-x) \\ y' &= y+r(q-y) \end{aligned} \right\} \text{ con } 0 \leq r \leq 1$$

S. EN DESLIZAMIENTO DE EJE $Ax+By+C=0$ y VECTOR (V,W):

Para que el eje y el vector sean paralelos debe verificarse $AV+BW=0$.

$$p = x - 2A \frac{Ax+By+C}{A^2+B^2} + V$$

$$q = y - 2B \frac{Ax+By+C}{A^2+B^2} + W$$

$$\left. \begin{aligned} x' &= x+r(p-x) \\ y' &= y+r(q-y) \end{aligned} \right\} \text{ con } 0 \leq r \leq 1$$

HOMOTECIA DE CENTRO (A,B) y RAZON K:

$$p = K(x-A) + A$$

$$q = K(y-B) + B$$

$$\left. \begin{aligned} x' &= x+r(p-x) \\ y' &= y+r(q-y) \end{aligned} \right\} \text{ con } 0 \leq r \leq 1$$

Al comparar estas fórmulas con las del programa se nota, además de algunas modificaciones de su forma, un cambio de signo en el ángulo de giro y en las ordenadas; esto se debe a que las ordenadas positivas de la pantalla coinciden con las ordenadas negativas usuales.

En este programa utilizamos algunas sentencias propias del Basic Applesoft (de los ordenadores Apple). Su función es la siguiente:

GET: acepta un carácter del teclado (o de un fichero). Su función es semejante a la de INPUT, salvo que el dato sólo puede tener un carácter de longitud.

HCOLOR=n: asigna un color al cursor de la página de alta resolución. Hemos usado $n=3$ (blanco) y $n=0$ (negro). En Apple II no hay ninguna sentencia que permita modificar el color de fondo de las pantallas gráficas.

HGR: pasa de la página de textos a la de gráficas de alta resolución; además borra el contenido actual de la página de gráficos.

HOME: borra la pantalla y sitúa el cursor en la posición superior izquierda.

HPlot x,y: dibuja el punto de coordenadas (x,y) de la página de alta resolución en el color previamente indicado.

HPlot x,y TO z,t: dibuja un segmento desde (x,y) hasta (z,t) en la página de alta resolución.

HTAB (m) y VTAB (n): marcan posiciones verticales (1 n 24) y horizontales (1 m 40) para situar el cursor en la pantalla.

TEXT: pasa de la página de gráficos a la de textos.

":": es el separador usado para escribir varias sentencias en la misma línea del programa.

La siguiente tabla refleja la equivalencia de algunas de estas sentencias para otros ordenadores:

APPLE	SPECTRUM	DRAGON 32	COMMODORE 64	ATARI	TI 99/4 A
GET	INKEY\$	INPUT# INKEY\$	GET(#)	GET#	CALL KEY
HCOLOR=n	INK n	PSET x,y,n COLOR n	-----	COLORn	-----
HOME	CLS	CLEAR	PRINT CHR\$()	GRAPHICS	CALL CLEAR
HPlot x,y	PLOT x,y	PSET x,y,n	-----	PLOT x,y	-----
HTAB(m) VTAB(n)	PRINT AT m,n	PRINT(@m	-----	-----	CALL HCHAR (n,m,...)
TEXT	-----	-----	-----	GRAPHICS	-----

La pantalla de gráficos de alta resolución del AppleII esta formada por 280 x 192 puntos. No obstante, existe también la posibilidad de mantener las cuatro líneas inferiores de texto en la pantalla, por lo que la zona gráfica se reduce a 280 x 160 puntos. Con el fin de utilizar el máximo posible de superficie dibujable, a lo largo del programa se pasa de una modalidad de pantalla a la otra, por medio de sentencias POKE:

POKE- 16302,0 elimina las cuatro líneas de texto y permite, usar la pantalla de gráficos de 280 x 192 puntos.

POKE- 16301,0 es la inversa de la anterior; reduce la pantalla de alta resolución a 280 x 160 puntos y deja visibles cuatro líneas del texto.

Para adaptar el programa a un ordenador que no disponga de estas posibilidades gráficas, sólo hay que eliminar las sentencias POKE anteriores.

Otra propiedad interesante del programa es que permite realizar varias transformaciones sucesivas de la misma figura (ver la instrucción 760). Como la selección de transformaciones no cabe en cuatro líneas de texto,

para elegir la segunda transformación de la figura hay que pasar la pantalla a texto y, después de hecha la elección, volver la pantalla a los gráficos. Si utilizáramos la sentencia HGR, se borraría el contenido de la página gráfica de la memoria del ordenador; este inconveniente se evita utilizando **POKE - 16304,0**, que permite el paso de texto a gráficos sin borrar nada. En un ordenador sin esta posibilidad, será necesario eliminar la sentencia **POKE - 16304,0** (línea 230) y modificar el menú de transformaciones para que quepa en la zona visible del texto.

```

10 TEXT : HOME
20 PRINT "*****"
30 PRINT "*** **"
40 PRINT "*** SEMEJANZAS EN EL PLANO ***"
50 PRINT "*** **"
60 PRINT "*** POR ***"
70 PRINT "*** **"
80 PRINT "*** ADELA JAIME Y ANGEL GUTIERREZ ***"
90 PRINT "*** **"
100 PRINT "*****"
110 VTAB (23): PRINT "PARA EMPEZAR PULSE 'RETURN' ";: GET A#
120 TEXT : HOME
130 VTAB (8): PRINT "SELECCIONE UNA TRANSFORMACION :": PRINT
140 PRINT TAB( 15);"0 SALIDA": PRINT
150 PRINT TAB( 15);"1 TRASLACION": PRINT
160 PRINT TAB( 15);"2 GIRO": PRINT
170 PRINT TAB( 15);"3 SIMETRIA": PRINT
190 PRINT TAB( 15);"4 SIM. EN DESLIZAMIENTO": PRINT
200 PRINT TAB( 15);"5 HOMOTECIA"
210 VTAB (21): INPUT S: PRINT : IF S = 0 THEN END
220 IF (S < 1) OR (S > 5) OR (S < > INT (S)) THEN PRINT CHR$ (7): GOTO
210
230 IF A# = "S" THEN POKE - 16304,0: GOTO 270: REM PASA DE TEXTO A GR
AFICA SIN BORRAR LAS FIGURAS
240 HGR : HCOLOR= 3
250 REM *****
260 REM DIBUJA LOS EJES DE COORDENADAS
270 POKE - 16302,0: REM PONE LA PANTALLA EN GRAFICA COMPLETA
280 HPLOT 0,95 TO 279,95
290 HPLOT 139,0 TO 139,191: HPLOT 138,0 TO 138,191
300 FOR I = - 13 TO 13
310 HPLOT 139 + I * 10,94 TO 139 + I * 10,96: HPLOT 138 + I * 10,94 TO 1
38 + I * 10,96
320 NEXT I
330 FOR I = - 9 TO 9
340 HPLOT 136,95 + 10 * I TO 141,95 + 10 * I
350 NEXT I
360 IF A# = "S" THEN GOTO 650
370 REM *****
380 REM DIBUJO DE LA FIGURA QUE SE VA A TRANSFORMAR
390 POKE - 16301,0: REM PONE LA PANTALLA EN TEXTO-GRAFICA
400 PRINT "DIBUJE UNA FIGURA DE HASTA 8 VERTICES"
410 I = 1: PRINT "ABCISA : -139<=X<=140": INPUT "X(1)= ";X
420 PRINT "ORDENADA : -96<=Y<=95": INPUT "Y(1)= ";Y: PRINT
430 X(1) = 139 + X:Y(1) = 95 - Y
440 M(1) = X(1):N(1) = Y(1)
450 POKE - 16302,0: HPLOT X(1),Y(1)
460 I = I + 1
470 IF I > 9 THEN PRINT : GOTO 640
480 POKE - 16301,0
490 IF I = 9 THEN GOTO 610
500 PRINT "ABCISA : -139<=X<=140"
510 PRINT "SI HA TERMINADO DE DIBUJAR, HAGA X=999"
520 PRINT "X(";I;")= ";: INPUT X
530 IF (X = 999) AND (I > 2) THEN PRINT : GOTO 640

```

```

540 IF X = 999 THEN X(I) = X(I - 1):Y(I) = Y(I - 1):M(I) = M(I - 1):N(I)
    = N(I - 1):I = I + 1: PRINT : GOTO 640
550 PRINT "ORDENADA : -96<=Y<=95"
560 PRINT "Y(";I;")= "; INPUT Y: PRINT
570 X(I) = 139 + X:Y(I) = 95 - Y
580 M(I) = X(I):N(I) = Y(I)
590 POKE - 16302,0: H PLOT X(I - 1),Y(I - 1) TO X(I),Y(I)
600 GOTO 460
610 PRINT : INPUT "YA NO QUEDAN VERTICES. QUIERE UNIR EL ULTIMO CON EL
    PRIMERO? (S/N) ";B$
620 IF LEFT$(B$,1) = "N" THEN PRINT : GOTO 640
630 X(I) = X(1):Y(I) = Y(1): GOTO 580
640 UV = I - 1: REM UV = ULTIMO VERTICE DE LA FIGURA
650 REM *****
660 POKE - 16301,0: ON S GOSUB 940,1070,1250,1520,1880
670 REM *****
680 REM VUELVE A DIBUJAR LA FIGURA INICIAL
690 FOR I = 2 TO UV
700 H PLOT X(I - 1),Y(I - 1) TO X(I),Y(I)
710 REM TOMA LA IMAGEN COMO FIGURA INICIAL
720 X(I - 1) = Z(I - 1):Y(I - 1) = T(I - 1)
730 NEXT I
740 X(UV) = Z(UV):Y(UV) = T(UV)
750 POKE - 16301,0: HOME : VTAB (21): INPUT "PARA CONTINUAR PULSE 'RETU
    RN'";B$: PRINT
760 INPUT "QUIERE TRANSFORMAR OTRA VEZ LA FIGURA? (S/N) ";A$:A$ = LEFT$(
    A$,1)
770 IF A$ = "N" THEN POKE - 16302,0: END
780 GOTO 120
790 REM *****
800 REM SUBRUTINAS
810 REM *****
820 REM DIBUJO DE LAS TRANSFORMACIONES
830 HCOLOR= 0: FOR I = 2 TO UV
840 H PLOT M(I - 1),N(I - 1) TO M(I),N(I)
850 NEXT I
860 HCOLOR= 3: FOR I = 2 TO UV
870 H PLOT Z(I - 1),T(I - 1) TO Z(I),T(I)
880 NEXT I
890 FOR I = 1 TO UV
900 M(I) = Z(I):N(I) = T(I)
910 NEXT I
920 RETURN
930 REM *****
940 REM TRASLACION
950 PRINT "EL VECTOR EMPIEZA EN (0,0) Y TERMINA EN (A,B), CON -139<=A<=1
    40 Y -96<=B<=95"
960 INPUT "A= ";A: INPUT "B= ";B
970 POKE - 16302,0
980 H PLOT 139,95 TO 139 + A,95 - B
990 P = 8: FOR M = 0 TO P: REM SE TRASLADA LA FIGURA EN P PASOS
1000 FOR I = 1 TO UV
1010 Z(I) = X(I) + A * M / P:T(I) = Y(I) - B * M / P
1020 NEXT I
1030 GOSUB 820
1040 NEXT M
1050 RETURN
1060 REM *****
1070 REM GIRO

```

```

1080 PRINT "EL CENTRO DE GIRO ES (A,B), CON          -138<=A<=130 Y -95<=
      B<=94"
1090 INPUT "A= ";A: INPUT "B= ";B:A = 139 + A:B = 95 - B
1100 POKE - 16302,0: H PLOT A - 1,B TO A + 1,B: H PLOT A,B - 1 TO A,B + 1

1110 POKE - 16301,0
1120 INPUT "ANGULO DE GIRO (ENTRE -360 Y +360 GR.) :";Z:Z = - Z: REM C
      AMBIO DE SIGNO DEBIDO A LAS COORDENADAS DE LA PANTALLA
1130 POKE - 16302,0
1140 P = 8: REM SE GIRA LA FIGURA EN P PASOS
1150 FOR M = 0 TO P
1160 R = (Z * M / P) * 3.1415926 / 180
1170 FOR I = 1 TO UV
1180 Z(I) = X(I) * COS (R) - Y(I) * SIN (R) + A * (1 - COS (R)) + B *
      SIN (R)
1190 T(I) = X(I) * SIN (R) + Y(I) * COS (R) + B * (1 - COS (R)) - A *
      SIN (R)
1200 NEXT I
1210 GOSUB 820
1220 NEXT M
1230 RETURN
1240 REM *****
1250 REM SIMETRIA
1260 PRINT "INDIQUE DOS PUNTOS DEL EJE DE SIMETRIA: "
1270 INPUT "-139<=X1<=140      X1= ";X1
1280 INPUT "-96<=Y1<=95      Y1= ";Y1
1290 INPUT "-139<=X2<=140      X2= ";X2
1300 INPUT "-96<=Y2<=95      Y2= ";Y2
1310 REM ECUACION DEL EJE DE SIMETRIA AX+BY+C=0
1320 POKE - 16302,0
1330 IF X1 = X2 THEN A = 1:B = 0:C = - X1: H PLOT 139 + X1,0 TO 139 + X1
      ,191: GOTO 1400: REM EJE VERTICAL
1340 A = Y2 - Y1:B = X1 - X2:C = X2 * Y1 - X1 * Y2
1350 FOR I = - 139 TO 140: REM DIBUJA EL EJE DE SIMETRIA
1360 Y = - (A * I + C) / B:X = 139 + I:Y = 95 - Y
1370 IF (X < 0) OR (X > 279) OR (Y < 0) OR (Y > 191) THEN GOTO 1390
1380 H PLOT X,Y
1390 NEXT I
1400 FOR I = 1 TO UV
1410 X = X(I) - 139:Y = 95 - Y(I)
1420 P(I) = X - 2 * A * (A * X + B * Y + C) / (A * A + B * B):Q(I) = Y -
      2 * B * (A * X + B * Y + C) / (A * A + B * B): REM (P(I),Q(I)) ES EL
      PUNTO SIMETRICO DE (X(I),Y(I))
1430 NEXT I
1440 P = 8: FOR M = 0 TO P: REM MUEVE LA FIGURA EN P PASOS
1450 FOR I = 1 TO UV
1460 Z(I) = X(I) + (139 + P(I) - X(I)) * M / P:T(I) = Y(I) + (95 - Q(I) -
      Y(I)) * M / P
1470 NEXT I
1480 GOSUB 820
1490 NEXT M
1500 RETURN
1510 REM *****
1520 REM SIMETRIA EN DESLIZAMIENTO
1530 PRINT "INDIQUE DOS PUNTOS DEL EJE DE SIMETRIA: "
1540 INPUT "-139<=X1<=140      X1= ";X1
1550 INPUT "-96<=Y1<=95      Y1= ";Y1
1560 INPUT "-139<=X2<=140      X2= ";X2
1570 INPUT "-96<=Y2<=95      Y2= ";Y2
1580 REM ECUACION DEL EJE DE SIMETRIA AX+BY+C=0
1590 POKE - 16302,0

```

```

1600 IF X1 = X2 THEN A = 1: B = 0: C = - X1: H PLOT 139 + X1, 0 TO 139 + X1
      , 191: GOTO 1670: REM EJE VERTICAL
1610 A = Y2 - Y1: B = X1 - X2: C = X2 * Y1 - X1 * Y2
1620 FOR I = - 139 TO 140: REM DIBUJA EL EJE DE SIMETRIA
1630 Y = - (A * I + C) / B: X = 139 + I: Y = 95 - Y
1640 IF (X < .0) OR (X > 279) OR (Y < 0) OR (Y > 191) THEN GOTO 1660
1650 H PLOT X, Y
1660 NEXT I
1670 POKE - 16301, 0
1680 PRINT "EL VECTOR EMPIEZA EN (0,0) Y TERMINA EN (V,W), CON -139<=V<=
      140, -96<=W<=95 Y"
1690 REM CONDICIONES DE PARALELISMO DEL EJE Y EL VECTOR
1700 IF B = 0 THEN PRINT "V=0": INPUT "W= "; W: V = 0: GOTO 1740
1710 IF A = 0 THEN PRINT "W=0": INPUT "V= "; V: W = 0: GOTO 1740
1720 PRINT "W="; - A / B; "V"
1730 INPUT "V= "; V: W = - A * V / B
1740 POKE - 16302, 0
1750 H PLOT 139, 95 TO 139 + V, 95 - W: H PLOT 138 + V, 95 - W TO 140 + V, 95 -
      W: H PLOT 139 + V, 94 - W TO 139 + V, 96 - W
1760 FOR I = 1 TO UV
1770 X = X(I) - 139: Y = 95 - Y(I)
1780 P(I) = X - 2 * A * (A * X + B * Y + C) / (A * A + B * B): Q(I) = Y -
      2 * B * (A * X + B * Y + C) / (A * A + B * B): REM (P(I), Q(I)) ES EL
      PUNTO SIMETRICO DE (X(I), Y(I))
1790 NEXT I
1800 P = 8: FOR M = 0 TO P: REM MUEVE LA FIGURA EN P PASOS
1810 FOR I = 1 TO UV
1820 Z(I) = X(I) + (139 + P(I) - X(I) + V) * M / P: T(I) = Y(I) + (95 - Q(
      I) - Y(I) - W) * M / P
1830 NEXT I
1840 GOSUB 820
1850 NEXT M
1860 RETURN
1870 REM *****
1880 REM HOMOTECIA
1890 PRINT "EL CENTRO DE HOMOTECIA ES (A,B), CON -139<A<140 Y -96<B<9
      5"
1900 INPUT "A= "; A: INPUT "B= "; B: A = 139 + A: B = 95 - B
1910 POKE - 16302, 0: H PLOT A - 1, B TO A + 1, B: H PLOT A, B - 1 TO A, B + 1

1920 POKE - 16301, 0: INPUT "RAZON DE HOMOTECIA : "; K
1930 POKE - 16302, 0
1940 P = 8: FOR M = 0 TO P: REM TRANSFORMA LA FIGURA EN P PASOS
1950 FOR I = 1 TO UV
1960 Z(I) = M / P * (K * (X(I) - A) + A) + (1 - M / P) * X(I): T(I) = M /
      P * (K * (Y(I) - B) + B) + (1 - M / P) * Y(I)
1970 NEXT I
1980 GOSUB 820
1990 NEXT M
2000 RETURN

```